
Wissenswertes über Textfunktionen

Mit Hilfe von Textfunktionen können Sie Text-Zeichenfolgen in Formeln manipulieren. Beispielsweise können Sie die Groß-/Kleinschreibung einer Text-Zeichenfolge ändern oder deren Länge ermitteln. Sie können auch ein Datum mit einer Text-Zeichenfolge verknüpfen bzw. verketten. Das folgende Beispiel zeigt eine Formel, in der die Funktion HEUTE zusammen mit der Funktion TEXT verwendet wird, um eine Meldung zu erzeugen, die das aktuelle Datum im Zahlenformat "tt-MMM-jj" anzeigt.

="Haushaltsplan vom  "&TEXT(HEUTE(),"tt-MMM-jj")

Zu welcher Funktion benötigen Sie weitere Informationen ?

SÄUBERN

Löscht alle nicht druckbaren Zeichen aus einem Text. Verwenden Sie SÄUBERN für Texte, die aus anderen Anwendungsprogrammen importiert wurden und eventuell Zeichen enthalten, die das von Ihnen verwendete Betriebssystem nicht drucken kann. Beispielsweise können Sie SÄUBERN dazu einsetzen, maschinennahen Code zu entfernen, der sich häufig am Anfang und Ende einer Datendatei befindet und nicht gedruckt werden kann.

Syntax

SÄUBERN(Text)

Text   ist eine beliebige Tabelleninformation, aus der Sie die nicht druckbaren Zeichen entfernen möchten.

Beispiel

Da die Funktion ZEICHEN(7) ein nichtdruckbares Zeichen liefert, gibt

SÄUBERN(ZEICHEN(7)&"Text"&ZEICHEN(7)) den Wert "Text" aus.

CODE

Liefert die Codezahl des ersten Zeichens in einem Text. Die ausgegebene Codezahl entspricht dem Zeichensatz, mit dem Ihr Rechner arbeitet.

Betriebssystem-Umgebung
Zeichensatz

Macintosh
Macintosh-Zeichensatz

Windows
ANSI

Syntax

CODE(Text)

Text   ist der Text, für den Sie die Codezahl des ersten Zeichens bestimmen möchten.

Beispiele

CODE("A") ergibt 65

CODE("Alphabet") ergibt 65

VERKETTEN

Verknüpft einzelne Textelemente zu einer Zeichenkette.

Syntax

VERKETTEN (Text1;Text2; ...)

Text1; Text2; ...   sind 1 bis 30 Argumente, die Zeichenfolgen (Texte) angeben, die zu einer Zeichenfolge verbunden werden sollen. Als Argumente dürfen Zeichenfolgen (Texte), Zahlen und Bezüge auf einzelne Zellen verwendet werden.

Hinweis

Anstelle von VERKETTEN kann der Operator "&" verwendet werden, um als Zeichenfolgen vorliegende Elemente miteinander zu verbinden.

Beispiele

VERKETTEN("Gesamt";"wert") ergibt "Gesamtwert". Dies ist identisch zu der Eingabe von "Gesamt"&"wert".

Angenommen innerhalb einer Tabelle zur Untersuchung eines Flusses enthält C2 "Bachforelle", C5 "Spezies" und C8 die Summe 32.

VERKETTEN("Die Flußpopulation der ";C5;" ";C2;"ist ";C8;"/Kilometer.") ergibt "Die Flußpopulation der Spezies Bachforelle ist 32/Kilometer."

DM

Wandelt eine Zahl in einen Text im Währungsformat um. Dabei werden die Dezimalstellen entsprechend der angegebenen Stelle rundet. Als Format wird #.##0,00 DM;-#.##0,00 DM verwendet.

Syntax

DM(Zahl;Dezimalstellen)

Zahl   ist eine Zahl, ein Bezug auf eine Zelle, die eine Zahl enthält, oder eine Formel, deren Ergebnis eine Zahl ist.

Dezimalstellen   ist die Anzahl der Ziffern rechts vom Dezimalkomma. Ist Dezimalstellen negativ, erfolgt das Runden von Zahl links vom Dezimalkomma. Fehlt das Argument Dezimalstellen, wird es als 2 angenommen.

Hinweis

Der Hauptunterschied zwischen dem Formatieren einer Zelle mit einer Zahl, die eine Zahl enthält, mit dem Befehl Zellen aus dem Menü Format und dem direkten Formatieren einer Zahl mit DM besteht darin, daß die Funktion DM ihr jeweiliges Ergebnis in Text umwandelt. Demgegenüber ist eine mit dem Befehl Zellen formatierte Zahl nach wie vor eine Zahl. Ungeachtet dieses Unterschieds können mit DM formatierte Zahlen in Formeln verwendet werden, da Microsoft Excel Zahlen, die als Textwerte eingegeben wurden, bei einem Rechenvorgang in normale Zahlen umwandelt.

Beispiele

DM(1234,567; 2) ergibt "1.234,57 DM"

DM(1234,567; -2) ergibt "1.200 DM"

DM(-1234,567; -2) ergibt "-1.200 DM"

DM(-0,123; 4) ergibt "-0,1230 DM"

DM(99,888) ergibt "99,89 DM"

IDENTISCH

Prüft, ob zwei Zeichenfolgen identisch sind. In diesem Fall wird WAHR zurückgegeben, ansonsten liefert die Funktion den Wert FALSCH. IDENTISCH unterscheidet nach Groß- und Kleinbuchstaben, ignoriert aber Formatierungsunterschiede. Verwenden Sie IDENTISCH, um Text zu prüfen, der in ein Dokument eingegeben wird.

Syntax

IDENTISCH(Text1;Text2)

Text1   ist die erste Zeichenfolge.

Text2   ist die zweite Zeichenfolge.

Beispiele

IDENTISCH("Wort"; "Wort") ergibt WAHR

IDENTISCH("Wort"; "wort") ergibt FALSCH

IDENTISCH("W ort"; "Wort") ergibt FALSCH

Sie können mit dieser Funktion beispielsweise prüfen, ob ein von einem Benutzer eingegebener Wert mit einem der Werte in einem festgelegten Bereich übereinstimmt. Geben Sie hierzu die folgende Formel als eine Matrix in eine Zelle ein. Um eine Formel als Matrix in eine einzelne Zelle einzugeben, drücken Sie STRG+UMSCHALT+EINGABE (in Microsoft Excel für Windows) oder  +EINGABE (in Microsoft Excel für den Macintosh). Der Name ZuPrüfWert bezieht sich auf die Zelle, die den jeweils vom Benutzer eingegebenen Wert enthält, und der Name AbgleichBereich bezieht sich auf die Liste der Textwerte, gegen die abgeglichen werden soll.

{=ODER(IDENTISCH(ZuPrüfWert; AbgleichBereich))}

FINDEN

Sucht eine Zeichenfolge innerhalb einer anderen (Groß-/Kleinschreibung wird beachtet) und liefert als Ergebnis die Nummer der Anfangsposition des Suchtextes, beginnend beim linken Zeichen des Textes. Sie können auch die Funktion SUCHEN verwenden, um eine Zeichenfolge innerhalb einer anderen zu suchen, aber im Gegensatz zur Funktion SUCHEN berücksichtigt die Funktion FINDEN die Schreibweise (Groß-/Kleinbuchstaben) der beteiligten Zeichenfolgen und läßt keine Platzhalterzeichen zu.

Syntax

FINDEN(Suchtext;Text;Erstes_Zeichen)

Suchtext   ist der Text, den Sie finden wollen.

·
Ist Suchtext eine leere Zeichenfolge (""), ermittelt FINDEN, ob die Übereinstimmung ab dem ersten Zeichen der zu durchsuchenden Zeichenfolge vorliegt (ab dem Zeichen, das die Position Erstes_Zeichen oder 1 einnimmt).

·
Suchtext darf keine Platzhalter enthalten.

Text   ist der Text, in dem Suchtext gesucht werden soll.

Erstes_Zeichen   gibt an, bei welchem Zeichen die Suche begonnen werden soll. Das erste zu Text gehörende Zeichen hat die Nummer 1. Fehlt das Argument Erstes_Zeichen, wird es als 1 angenommen.

Hinweis

·
Wird Suchtext nicht in Text gefunden, liefert FINDEN den Fehlerwert #WERT!.

·
Ist Erstes_Zeichen kleiner gleich 0, liefert FINDEN den Fehlerwert #WERT!.

·
Ist Erstes_Zeichen größer als die Länge von Text, liefert FINDEN den Fehlerwert #WERT!.

Beispiele

FINDEN("M";"Miriam Meier") ergibt 1

FINDEN("m";"Miriam Meier") ergibt 6

FINDEN("M";"Miriam Meier";3) ergibt 8

Eine Tabelle enthält z.B. eine Liste mit Teilenamen und -nummern, und Sie möchten aus den entsprechenden Zellen nur die Namen der Teile lesen, nicht deren Nummern. Sie können mit der Funktion FINDEN nach dem Nummernzeichen (#) suchen und die Funktion TEIL dazu verwenden, die Teilenummern unberücksichtigt zu lassen. Die Zellen A2:A4 enthalten die folgenden Einträge (Teilenamen und -nummern): "Keramischer Isolator #124-TD45-87", "Kupferspule #12-671-6772", "Regelwiderstand #116010".

TEIL(A2;1;FINDEN(" #";A2;1)-1) liefert "Keramischer Isolator"

TEIL(A3;1;FINDEN(" #";A3;1)-1) liefert "Kupferspule"

TEIL(A4;1;FINDEN(" #";A4;1)-1) liefert "Regelwiderstand"

FEST

Formatiert eine Zahl als Text mit einer festen Anzahl an Nachkommastellen.

Syntax

FEST(Zahl;Dezimalstellen;Keine_Punkte)

Zahl   ist die Zahl, die Sie runden und in Text umwandeln möchten.

Dezimalstellen   ist die Anzahl der Ziffern rechts vom Dezimalkomma.

Keine_Punkte   ist ein Wahrheitswert. Ist Keine_Punkte WAHR, fügt FEST in einen als Ergebnis gelieferten Text keine Punkte (1.000er-Trennzeichen) ein. Ist Keine_Punkte FALSCH oder nicht angegeben, enthält der zurückgegebene Text die üblicherweise verwendeten Punkte.

·
In Microsoft Excel können Zahlen maximal 15 gültige Ziffern haben, aber der größte zulässige Wert für Dezimalstellen ist 127.

·
Ist Dezimalstellen negativ, werden die Stellen von Zahl links vom Dezimalkomma gerundet.

·
Fehlt das Argument Dezimalstellen, wird es als 2 angenommen.

Hinweis

Der Unterschied zwischen dem Formatieren einer Zahl in einer Zelle mit dem Befehl Zellen aus dem Menü Format und dem direkten Formatieren einer Zahl mit der Funktion FEST besteht darin, daß die Funktion FEST ihr jeweiliges Ergebnis in Text umwandelt. Demgegenüber ist eine mit dem Befehl Zellen formatierte Zahl nach wie vor eine Zahl.

Beispiele

FEST(1234,567; 1) ergibt "1.234,6"

FEST(1234,567; -1) ergibt "1.230"

FEST(-1234,567; -1) ergibt "-1.230"

FEST(44,332) ergibt "44,33"

LINKS

Diese Funktion liefert äußeren linken Zeichen einer Zeichenfolge.

Syntax 

LINKS(Text;Anzahl_Zeichen)

Text   ist die Zeichenfolge, die die Zeichen umfaßt, die Sie herauskopieren möchten.

Anzahl_Zeichen   gibt an, wie viele Zeichen LINKS liefern soll.

·
Anzahl_Zeichen muß größer gleich null sein.

·
Gibt Anzahl_Zeichen mehr Zeichen an, als der Text lang ist, liefert LINKS den gesamten Text.

·
Fehlt das Argument Anzahl_Zeichen, wird es als 1 angenommen.

Beispiele

LINKS("Verkaufspreis"; 7) ergibt "Verkauf"

Enthält A1 die Zeichenfolge "Schweden", liefert

LINKS(A1) den Wert "S"

Einige Buchhaltungsprogramme zeigen rechts neben einem negativen Werte das Minuszeichen an (–). Wenn Sie eine Datei importieren, die in einem Programm erstellt wurde, das negative Werte auf diese Weise speichert, importiert Microsoft Excel diese Werte unter Umständen als Text. Um die Textzeichenfolgen in Werte umzuwandeln, müssen Sie alle Zeichen der Textzeichenfolge zurückgeben, mit Ausnahme des Minuszeichens, und das Ergebnis dann mit –1 multiplizieren. Verwenden Sie für das Argument Anzahl_Zeichen die Tabellenfunktion LÄNGE, um die Anzahl der Zeichen in der Zeichenfolge zu bestimmen, und subtrahieren Sie dann 1. Ist beispielsweise der Wert in Zelle A2 "156–", wandelt die folgende Formel den Text in den Wert –156 um.

LINKS(A2,LÄNGE(A2)–1)*–1

LÄNGE

Liefert die Anzahl der Zeichen einer Zeichenfolge.

Syntax 

LÄNGE(Text)

Text   ist der Text, dessen Länge Sie ermitteln möchten. Leerzeichen zählen als Zeichen.

Beispiele

LÄNGE("Karlsruhe, BW") ergibt 13

LÄNGE("") ergibt 0

KLEIN

Wandelt einen Text in Kleinbuchstaben um.

Syntax 

KLEIN(Text)

Text   ist der Text, den Sie in Kleinbuchstaben umwandeln möchten. KLEIN nimmt an Zeichen des Textes, die keine Buchstaben sind, keine Änderungen vor.

Beispiele

KLEIN("M. S. Wetter") ergibt "m. s. wetter"

KLEIN("Abt. 2B") ergibt "abt. 2b"

KLEIN ist verwandt mit GROSS2 und GROSS. Siehe auch Beispiele zu GROSS2.

TEIL

Liefert eine bestimmte Anzahl Zeichen einer Zeichenfolge ab der von Ihnen bestimmten Stelle.

Syntax

TEIL(Text;Erstes_Zeichen;Anzahl_Zeichen)

Text   ist die Zeichenfolge, die die Zeichen umfaßt, die Sie herauskopieren möchten.

Erstes_Zeichen   ist die Position des ersten Zeichens, das Sie aus dem Text herauskopieren möchten. Für das erste Zeichen von Text gilt, daß Erstes_Zeichen den Wert 1 hat, und so weiter.

·
Ist Erstes_Zeichen größer als die Gesamtlänge von Text, liefert TEIL die Zeichenfolge "" (leere Zeichenfolge).

·
Ist Erstes_Zeichen kleiner als die Gesamtlänge von Text, aber Erstes_Zeichen plus Anzahl_Zeichen übersteigt die Gesamtlänge von Text, liefert TEIL alle bis zum Ende von Text gehörenden Zeichen.

·
Ist Erstes_Zeichen kleiner als 1, liefert TEIL den Fehlerwert #WERT!.

Anzahl_Zeichen   gibt an, wie viele der zu Text gehörenden Zeichen herauskopiert werden sollen. Ist Anzahl_Zeichen negativ, liefert TEIL den Fehlerwert #WERT!.

Beispiele

TEIL("Wonnemonat Mai"; 1; 10) ergibt "Wonnemonat"

TEIL("Wonnemonat Mai"; 12; 20) ergibt "Mai"

TEIL("1234"; 5; 5) ergibt "" (leere Zeichenfolge)

Weitere Beispiele finden Sie bei den Funktionen CODE und FINDEN.

GROSS2

Wandelt den ersten Buchstaben aller Wörter einer Zeichenfolge in Großbuchstaben um. Wandelt alle anderen Buchstaben in Kleinbuchstaben um.

Syntax

GROSS2(Text)

Text   ist ein in Anführungszeichen eingeschlossener Text, eine Formel, die einen Text liefert, oder ein Bezug auf eine Zelle, die den Text enthält, den Sie teilweise großschreiben wollen.

Beispiele

GROSS2("dies ist ein TITEL") ergibt "Dies Ist Ein Titel"

GROSS2("10mal soviel") ergibt "10Mal Soviel"

GROSS2("76BudGet") ergibt "76Budget"

ERSETZEN

Ersetzt eine bestimmte Anzahl Zeichen ab einer bestimmten Stelle innerhalb eines Textes.

Syntax

ERSETZEN(Alter_Text;Erstes_Zeichen;Anzahl_Zeichen;Neuer_Text)

Alter_Text   ist der Text (die Zeichenfolge), in dem (der) Sie einige Zeichen ersetzen möchten.

Erstes_Zeichen   ist die Position des zu Alter_Text gehörenden Zeichens, an der mit dem Ersetzen durch Neuer_Text begonnen werden soll.

Anzahl_Zeichen   gibt an, wie viele der zu Alter_Text gehörenden Zeichen Sie durch Neuer_Text ersetzen möchten.

Neuer_Text   ist der Text (die Zeichenfolge), durch den (die) die zu Alter_Text gehörenden Zeichen ersetzt werden.

Beispiele

Die folgende Formel ersetzt fünf Zeichen von Neuer_Text ab dessen sechsten Zeichen durch Alter_Text:

ERSETZEN("abcdefghijk"; 6; 5; "*") ergibt "abcde*k"

Ab dem sechsten bis einschließlich des zehnten sind alle Zeichen durch "*" ersetzt.

Die folgende Formel ersetzt die beiden letzten Ziffern von 1990 durch 91:

ERSETZEN("1990"; 3; 2; "91") ergibt "1991"

Enthält die Zelle A2 die Zeichenfolge "123456", liefert

ERSETZEN(A2; 1; 3; "@") die Zeichenfolge "@456"

Gibt die Funktion RECHTS die Zeichenfolge "ABCDEF" zurück, liefert

ERSETZEN(RECHTS(A3; 6); 1; 6; "*") die Zeichenfolge "*"

WIEDERHOLEN

Wiederholt einen Text so oft wie angegeben. Verwenden Sie WIEDERHOLEN, um eine Zeichenfolge (eine Basiszeichenfolge) in einer bestimmten Häufigkeit in eine Zelle einzugeben.

Syntax

WIEDERHOLEN(Text;Multiplikator)

Text   ist der Text (die Basiszeichenfolge), den (die) Sie wiederholen möchten.

Multiplikator   ist eine positive Zahl, die angibt, wie oft Text wiederholt werden soll. Ist Multiplikator mit 0 (Null) belegt, liefert WIEDERHOLEN eine leere Zeichenfolge (""). Ist Multiplikator keine ganze Zahl, werden deren Nachkommastellen abgeschnitten. Eine von WIEDERHOLEN gelieferte Zeichenfolge kann aus höchstens 32.676 Zeichen bestehen.

Tip   Mit dieser Funktion können Sie innerhalb einer Tabelle ein einfaches Histogramm erstellen.

Beispiele

WIEDERHOLEN("*-"; 3) ergibt "*-*-*-"

Enthält die Zelle A3 die Zeichenfolge "Umsatz", liefert 

WIEDERHOLEN($A$3; 2,9) die Zeichenfolge "UmsatzUmsatz"

RECHTS

Liefert die äußeren rechten Zeichen einer Zeichenfolge.

Syntax

RECHTS(Text;Anzahl_Zeichen)

Text   ist die Zeichenfolge mit den Zeichen, die Sie kopieren möchten.

Anzahl_Zeichen   gibt an, wie viele Zeichen kopiert werden sollen.

·
Anzahl_Zeichen muß größer gleich 0 sein.

·
Gibt Anzahl_Zeichen mehr Zeichen an, als der Text lang ist, liefert RECHTS den gesamten Text.

·
Fehlt das Argument Anzahl_Zeichen, wird es als 1 angenommen. 

Beispiele

RECHTS("Verkaufspreis"; 5) ergibt "preis"

RECHTS("Hausnummer") ergibt "r"

RECHTS ist das Gegenstück zu LINKS; weitere Beispiele finden Sie unter der Funktion LINKS.

SUCHEN

Sucht eine Zeichenfolge innerhalb einer anderen (Groß-/Kleinschreibung wird nicht beachtet), und liefert die Nummer des Zeichens, ab dem die zu suchende Zeichenfolge erstmals gefunden wurde. SUCHEN können Sie z. B. immer dann einsetzen, wenn Sie ermitteln möchten, welche Position ein Zeichen oder eine Zeichenfolge innerhalb einer anderen Zeichenfolge einnimmt, um letztere mit einer der Funktionen TEIL oder ERSETZEN zu ändern.

Syntax

SUCHEN(Suchtext;Text;Erstes_Zeichen)

Suchtext   ist der Text, den Sie finden wollen. Eine als Suchtext angegebene Zeichenfolge darf die Platzhalterzeichen Fragezeichen (?) und Sternchen (*) enthalten. Ein Fragezeichen ersetzt ein Zeichen; ein Sternchen ersetzt eine beliebige Zeichenfolge. Suchen Sie nach einem Fragezeichen oder Sternchen, müssen Sie eine Tilde (~) vor das zu suchende Zeichen setzen. Wird die von Suchtext angegebene Zeichenfolge nicht gefunden, liefert die Funktion den Fehlerwert #WERT!.

Text   ist der Text, in dem Sie nach Suchtext suchen möchten

Erstes_Zeichen   ist die Nummer des Zeichens in Text (von links nach rechts), ab der Sie mit der Suche beginnen wollen.

·
Fehlt das Argument Erstes_Zeichen, wird es als 1 angenommen.

·
Ist Erstes_Zeichen mit einem Wert belegt, der kleiner als 1 oder größer als die Länge von Text ist, liefert die Funktion den Fehlerwert #WERT!.

Tip   Mit Erstes_Zeichen können Sie dafür sorgen, daß ein Suchvorgang nicht mit dem ersten Zeichen der zu durchsuchenden Zeichenfolge beginnt, sondern eine bestimmte Anzahl an Zeichen überspringt. Sie möchten z. B. die Zeichenfolge "MOD0093.JungeMännerMode" auswerten. Um festzustellen, welche Position das erste J im angegebenen Teil der Zeichenfolge einnimmt, müssen Sie Erstes_Zeichen mit 8 belegen, so daß der Abschnitt der Zeichenfolge, der die Seriennummer angibt, nicht durchsucht wird. SUCHEN beginnt mit dem achten Zeichen, stellt fest, daß das nächste Zeichen mit dem Suchtext übereinstimmt, und liefert die Zahl 9. SUCHEN liefert immer die Anzahl der Zeichen ab dem Anfang der ausgewerteten Zeichenfolge, nicht ab Erstes_Zeichen.

Hinweis

·
SUCHEN unterscheidet bei der Suche nach einer Zeichenfolge nicht zwischen Groß- und Kleinbuchstaben.

·
SUCHEN und FINDEN unterscheiden sich nur darin, daß FINDEN die Schreibweise berücksichtigt, also zwischen Groß- und Kleinbuchstaben unterscheidet.

Beispiele

SUCHEN("e";"Anweisungen";6) ergibt 10

Enthält die Zelle B17 das Wort "Spanne" und die Zelle A14 die Zeichenfolge "Gewinnspanne", liefert

SUCHEN($B$17;$A$14) den Wert 7

Verwenden Sie SUCHEN in Verbindung mit der Funktion ERSETZEN, um den korrekten Wert für  Erstes_Zeichen an Ersetzen zu übergeben und sicherzustellen, daß der neue Text an der richtigen Position eingefügt wird. Werden wieder die Zellbezüge des vorherigen Beispiels verwendet, liefert

ERSETZEN($A$14;SUCHEN($B$17;$A$14);6;"betrag") die Zeichenfolge "Gewinnbetrag".

WECHSELN

Tauscht einen alten Text durch einen neuen Text in einer Zeichenfolge aus. WECHSELN können Sie immer dann einsetzen, wenn Sie innerhalb eines Textes eine bestimmte Zeichenfolge austauschen möchten. ERSETZEN sollten Sie immer dann einsetzen, wenn Sie innerhalb eines Textes eine an einer bestimmten Position beginnende Zeichenfolge ersetzen möchten.

Syntax

WECHSELN(Text;Alter_Text;Neuer_Text;ntes_Auftreten)

Text   ist der Text oder der Bezug auf eine Zelle, die den Text enthält, in dem Sie Zeichen ersetzen möchten..

Alter_Text   ist die Zeichenfolge, die Sie ersetzen (wechseln) möchten.

Neuer_Text   ist die Zeichenfolge, durch die Sie Alter_Text ersetzen möchten.

ntes_Auftreten   gibt an, das wievielte Vorkommen von Alter_Text Sie durch Neuer_Text ersetzen möchten. Wenn Sie ntes_Auftreten angeben, wird nur dieses Vorkommen von Alter_Text ersetzt; andernfalls wird Alter_Text an jeder Stelle, an der er in Text vorkommt, durch Neuer_Text ersetzt.

Beispiele

WECHSELN("Anlageobjekt"; "objekt"; "gut") ergibt "Anlagegut"

WECHSELN("Quartal 1, 1991"; "1"; "2"; 1) ergibt "Quartal 2, 1991"

WECHSELN("Quartal 1, 1991"; "1"; "2"; 3) ergibt "Quartal 1, 1992"

Mit der folgenden Formel wird die Textkonstante namens "Trennzeichen" an jeder Stelle, an der sie innerhalb der Zelle "Umgebung2" vorkommt, durch eckige Klammern ersetzt:

WECHSELN(Umgebung2; Trennzeichen; "] [")

ZEICHEN

Liefert das der Codezahl entsprechende Zeichen. Verwenden Sie ZEICHEN, um Zahlen in einer anderen Kodierung, die Sie aus Dateien erhalten, die auf Rechnern anderen Typs erstellt wurden, in Zeichen umzuwandeln.

Systemumgebung
Zeichensatz

Macintosh
Macintosh-Zeichensatz

Windows
ANSI

Syntax

ZEICHEN(Zahl)

Zahl   ist eine Zahl von 1 bis 255, die das von Ihnen gewünschte Zeichen angibt. Das jeweilige Zeichen ist Bestandteil des Zeichensatzes, der auf Ihrem Rechner eingesetzt wird.

Beispiele

ZEICHEN(65) ergibt "A"

ZEICHEN(33) ergibt "!"

TEXT

Formatiert eine Zahl und wandelt sie in einen Text um.

Syntax

TEXT(Wert;Textformat)

Wert   ist ein numerischer Wert, eine Formel, die einen numerischen Wert liefert, oder ein Bezug auf eine Zelle, die einen numerischen Wert enthält.

Textformat   ist ein als Zeichenfolge angegebenes Zahlenformat im Feld Kategorie in der Registerkarte Zahlen im Dialogfeld Zellen. Textformat darf weder Sternchen (*) enthalten noch mit dem Numerierungsformat "Standard" belegt sein.

Hinweis

Wenn Sie eine Zelle mit einer Option in der Registerkarte Zahlen (Befehl Zellen, Menü Format) formatieren, wird nur das Format, nicht aber der Wert geändert. Wenn Sie die Funktion TEXT verwenden, wird der jeweilige Wert in einen formatierten Text umgewandelt, und das Ergebnis wird nicht länger als Zahl berechnet.

Beispiele

TEXT(2,715; "0,00 DM") ergibt "2,72 DM"

TEXT("15.4.91"; "T. MMMM JJJJ") ergibt "15. April 1991"

GLÄTTEN

Löscht Leerzeichen in einem Text, die nicht als jeweils einzelne zwischen Wörtern stehende Trennzeichen dienen. GLÄTTEN können Sie für Texte verwenden, die Sie aus anderen Anwendungsprogrammen übernommen haben und die eventuell unerwünschte Leerzeichen enthalten.

Syntax

GLÄTTEN(Text) 

Text   ist der Text, aus dem Sie Leerzeichen entfernen möchten.

Beispiel

GLÄTTEN(" Gewinn   im   ersten   Quartal   ") ergibt "Gewinn im ersten Quartal"

GROSS

Wandelt einen Text in Großbuchstaben um.

Syntax

GROSS(Text)

Text   ist der Text, der in Großbuchstaben umgewandelt werden soll.  Text kann sowohl ein Bezug als auch eine Zeichenfolge sein.

Beispiele

GROSS("total") ergibt "TOTAL"

Enthält die Zelle E5 die Zeichenfolge "Rendite", dann folgt:

GROSS(E5) ergibt "RENDITE"

WERT

Wandelt ein als Text angegebenes Argument in eine Zahl um.

Syntax

WERT(Text)

Text   gibt die Zeichenfolge an, die Sie umwandeln möchten, und steht entweder in Anführungszeichen oder ist ein Bezug auf die Zelle, die den Text enthält. Text kann in jedem der Formate vorliegen, die Microsoft Excel bezüglich konstanter Zahlen, Tagesdaten und Uhrzeiten unterstützt. Liegt Text in keinem dieser Formate vor, liefert die Funktion WERT den Fehlerwert #WERT!.

Hinweis

Normalerweise ist es nicht erforderlich, die Funktion WERT innerhalb einer Formel einzusetzen, da Microsoft Excel Texte bei Bedarf automatisch in Zahlen umwandelt. Diese Funktion steht aus Gründen der Kompatibilität zu anderen Tabellenkalkulationsprogrammen zur Verfügung.

Beispiele

WERT("1.000 DM") ergibt 1.000

WERT("16:48:00")-WERT("12:00:00") ist gleich "16:48:00"-"12:00:00" und ergibt 0,2, also die fortlaufende Zahl, die der Zeitspanne 4 Stunden und 48 Minuten entspricht.

T

Wandelt die Argumente in Text um.

Syntax

T(Wert)

Wert   ist der Wert, den Sie prüfen möchten. Ist Wert ein Text oder bezieht sich Wert auf einen Text, liefert T den tatsächlichen Wert. Bezieht sich Wert nicht auf einen Text, liefert T eine leere Zeichenfolge ("").

Hinweis

Normalerweise ist es nicht erforderlich, die Funktion T innerhalb einer Formel einzusetzen, da Microsoft Excel Werte bei Bedarf automatisch umwandelt. Diese Funktion steht aus Gründen der Kompatibilität zu anderen Tabellenkalkulationsprogrammen zur Verfügung.

Beispiele

Enthält B1 den Text "Niederschlag", liefert

T(B1) die Zeichenfolge "Niederschlag"

Enthält B2 die Zahl 19, liefert

T(B2) den Wert ""

T("Wahr") ergibt "Wahr"

T(WAHR) ergibt ""

